

Лабораторная работа №1

РЕШЕНИЕ СТАЦИОНАРНОЙ ЗАДАЧИ ТЕПЛОПРОВОДНОСТИ В ПЛОСКОЙ ОБЛАСТИ

Индивидуальные задания – тела в форме букв.

КЛЮЧЕВЫЕ СЛОВА

1. Задача теплопроводности (температурный анализ)
2. Статический температурный анализ
3. Плоская задача

ОПИСАНИЕ ЗАДАЧИ

В качестве примера рассмотрим стационарную задачу теплопроводности для двумерной области в виде латинской буквы «R». Геометрия области, отнесенной к декартовой системе координат Oxy , показана на рис. 1. Размеры буквы определяются следующими параметрами: $HL=1.4$ м (высота); $WL=0.8$ м (ширина); $WWL=0.2$ м (ширина стенок); $HAL=0.7$ м (дополнительный размер по высоте). Материалом области является сталь с коэффициентом теплопроводности $k=46.7$ (Вт/(м град)).




Рис. 1. Схема области

В задаче требуется определить поле температур и вектор потока тепла в плоской области в форме буквы «R» при подаче на нижние грани заданных значений температуры и при задании на

верхней грани условия конвективного теплообмена. Именно, предположим, что на левой нижней грани $0 \leq x \leq WWL$, $y = 0$, температура θ_h равна 90°C ; а на правой нижней грани $WL - WWL \leq x \leq WL$, $y = 0$, температура θ_h равна 180°C . На верхней грани $0 \leq x \leq WWL$, $y = HL$, имеем условие конвективного теплообмена со значениями коэффициента теплообмена 60 м^{-1} и температурой внешней среды 0°C . Все остальные границы считаются теплоизолированными.

ВВОДНЫЕ ЗАМЕЧАНИЯ

Заметим, что при задании входных параметров задачи нужно заботиться об их согласованности по системам единиц. Здесь выбрана система единиц – СИ, и, таким образом, геометрические размеры даются в метрах.

Для построения области в виде буквы «R», отметим, что ее правая верхняя часть ограничена кривыми линиями, которые можно задавать отрезками эллипсов. Центр эллипсов будет совпадать с центром декартовой системы координат $O'x'y'$, сдвинутой относительно исходной системы координат Oxy на WWL вправо по оси абсцисс, и на $(HL - HAL/2)$ вверх по оси ординат. Для внешней криволинейной части отношение полуосей эллипса по осям y' и x' равно: $=(HAL/2)/(WL - WWL)$.




Рис. 2. Схема области, нумерация областей и точек

Решение задачи с использованием ANSYS (APDL)

Для решения плоской стационарной задачи теплопроводности выберем шестиузловой треугольный температурный конечный элемент PLANE35 пакета ANSYS.

Будем строить сначала твердотельную модель, а затем, из нее – конечно-элементную. Твердотельную модель получим методом построения геометрических объектов «снизу – вверх». Для этого сначала построим опорные точки, потом – линии, и, наконец, двумерные области.

Нам понадобится сначала построить две односвязные области: первую область, ограниченную внешними линиями буквы «R», и вторую область, ограниченную внутренними линиями. Затем, с использованием булевой операции вычитания геометрических областей (команда **ASBA**) можно будет получить итоговую область буквы «R» с отверстием.

В соответствие с этой идеологией сначала, введя заголовок **/TITLE** (не обязательная процедура) и перейдя в препроцессор **/PREP7**, зададим идентификаторы:

```
/TITLE, Heat Flow in 2D region (R)  
/PREP7
```

! Все величины в системе Си

! Геометрические размеры тела в форме буквы "R"

HL=1.4 ! Высота

WL=0.8 ! Ширина

WWL=0.2 ! Ширина стенок буквы "R"

HAL=0.7 ! Дополнительный размер по высоте

T_INP1=90 ! Температура на левой части нижней границе

T_INP2=180 ! Температура на правой части нижней границе

T_EXT=0 ! Температура окружающей среды для конвективных условий

T_EXT=0

H_F=60 ! Коэффициент конвективного теплообмена

Далее, определим материальные свойства. Для линейной стационарной задачи теплопроводности для изотропной однородной среды нужно только одно значение: коэффициент теплопроводности $K=K_{XX}$. Значение K_{XX} для стали поместим в набор материальных свойств под номером «1» по команде:

```
MP,KXX,1,46.7 ! Коэффициент теплопроводности KXX=46.7 (Сталь)
```

Зададим параметры триангуляции. Пусть есть один глобальный размер DMESH для конечных элементов, равный половине ширины стенок буквы (т.е. $DMESH=WWL/2$), и более мелкий размер

DMESH1, в четыре раза меньший. Для определения этих величин введем соответствующие параметры:

! Параметры для триангуляции

DMESH=WWL/2

DMESH1=DMESH/4

Определим тип конечного элемента. Мы хотим разбить нашу область на конечные элементы одного типа: треугольные квадратичные температурные конечные элементы с шестью узлами PLANE35. Для этого поместим в первый номер типов элементов элемент PLANE35 по команде:

ET,1,PLANE35 ! Шестиузловой треугольный температурный КЭ PLANE35

Анализируя вид буквы «R», замечаем, что ее правая верхняя часть ограничена кривыми линиями, которые можно задавать отрезками эллипсов. Центр эллипсов будет совпадать с центром декартовой системы координат $O'x'y'$, сдвинутой относительно исходной системы координат Oxy на WWL вправо по оси абсцисс, и на $(HL-HAL/2)$ вверх по оси ординат. Для внешней криволинейной части отношение полуосей эллипса по осям y' и x' равно: $= (HAL/2)/(WL-WWL)$. Поэтому, для дальнейшего построения криволинейных линий удобно определить новую эллиптическую систему координат под номером 11, введя команды:

PEL=(HAL/2)/(WL-WWL) ! Параметр эллиптичности для криволинейной части фигуры

LOCAL,11,1,WWL,HL-HAL/2,,,,,PEL ! Эллиптическая система координат 11

CSYS,0 ! Переход в основную декартову систему координат

Теперь создаем область, ограниченную внешними линиями буквы «R». Сначала создаем девять опорных точек:

! Определение основных опорных точек для внешней границы

K,1,0,0

K,2,WWL,0

K,3,WWL,HL-HAL-WWL

K,4,WL-WWL,0

K,5,WL,0

K,6,WWL,HL-HAL

K,7,WL,HL-HAL/2

K,8,WWL,HL

K,9,0,HL

Затем определяем линии по точкам. При этом для создания эллиптической линии просто переходим в эллиптическую систему координат, и в ней проводим «прямую» линию, которая в результате

оказывается эллиптической. Команда **AL** в итоге позволяет создать нужную нам первую область A1:

! создание линий по точкам

L,1,2 \$ L,2,3 \$ L,3,4 \$ L,4,5 \$ L,5,6

CSYS,11 ! Переход в систему координат с номером 11

L,6,7 \$ L,7,8

CSYS,0 ! Переход в основную декартову систему координат

L,8,9 \$ L,9,1

AL,1,2,3,4,5,6,7,8,9 ! Определение площади 1 по линиям

Замечание. Область A1 можно было создать и по команде **A,1,2,3,4,5,6,7,8,9 !** Определение площади 1 по опорным точкам. Действительно, эта команда создает площадь (поверхность) по опорным точкам с проведением прямых линий между точками. Однако если между точками уже определены линии, то команда **A** использует их, и в результате оказывается эквивалентной команде **AL**.

Создавая другие области, нужно иметь в виду, что команда **AL** в командном режиме строит область максимум по десяти линиям, а команда **A** – максимум по восемнадцати точкам (если не используется значение **ALL** для первого поля). В интерактивном графическом режиме таких ограничений нет. Но если есть желание работать только в командном режиме с использованием **APDL**, то можно разбить сложную область на более простые подчасти с числом точек или линий, удовлетворяющих ограничениям команд **A** или **AL**. Создав эти подобласти, нужную сложную область можно получить объединением подобластей. Можно их и не объединять в одну область, но тогда нужно проследить, чтобы граничащие подобласти имели одни и те же опорные точки и линии на соседствующих границах.

Действуя аналогично, построим вторую область – пустотелую часть буквы «R»:

! Определение отверстия - площади 2

K,10,WWL,HL-HAL/2-(WL-2*WWL)*PEL

K,11,WL-WWL,HL-HAL/2

K,12,WWL,HL-HAL/2+(WL-2*WWL)*PEL

CSYS,11

L,10,11 \$ L,11,12

CSYS,0

L,12,10

AL,10,11,12

Теперь булевская операция вычитание позволяет из первой области вырезать вторую, и в результате получить требуемую область с отверстием, причем эта область будет иметь номер 3:

ASBA,1,2 ! Вырезать из области 1 область 2

В итоге создана твердотельная модель без граничных условий. Область под номером «3» соответствует планируемому виду буквы «R», и ей по умолчанию сопоставлены все атрибуты с номером «1», т.е. набор материальных свойств с номером «1», типы элементов с номером «1», и т.д. Команды **MP** и **ET**, введенные ранее, задали нужные нам значения для материальных свойств с номером «1» и типов элементов с номером «1».

Переход от твердотельной модели к конечно-элементной оказывается здесь совсем не сложным. Определив размеры конечных элементов вблизи опорных точек по командам **KESIZE**, получаем конечно-элементную модель из треугольных элементов PLANE35 по команде **AMESH**:

! Установка размеров элементов вблизи опорных точек

KESIZE,ALL,DMESH

KESIZE,3,DMESH1 \$ KESIZE,6,DMESH1

KESIZE,10,DMESH1 \$ KESIZE,11,DMESH1 \$ KESIZE,12,DMESH1

AMESH,ALL ! Триангулировать область (3)

FINISH

В конце приведенного выше фрагмента по команде **FINISH** была завершена работа постпроцессора.

Полученная конечно-элементная модель показана на рис. 3 (команда **EPLOT** при некоторых опциях графического вывода, приведенных далее в постпроцессоре). Для вывода конечно-элементной сетки в интерактивном режиме следует выполнить: Plot->Elements.




Рис. 3. Конечно-элементная модель

Следующим этапом является задание граничных условий и формирование и решение системы линейных алгебраических уравнений (СЛАУ) МКЭ. Все эти действия выполним в решателе **/SOLU**.

Для задания граничных условий мы выделяет нужное множество узлов командами **NSEL**, а затем главные граничные условия для температуры задаем по команде **D**, а условия конвективного теплообмена – по команде **SF**. По окончании этих процедур нужно не забыть вернуться к выбору всех узлов модели по команде **NSEL,ALL**.

По одной команде **SOLVE** происходит формирование элементных матриц, формирование СЛАУ МКЭ и собственно решение СЛАУ.

Блок нужных команд решателя в итоге выглядит так:

```
/SOLU  
ANTYPE,STAT ! Решение стационарной задачи  
NSEL,S,LOC,Y,0 ! Выбор всех узлов с координатой Y=0  
NSEL,R,LOC,X,0,WWL  
D,ALL,TEMP,T_INP1 ! Задать для всех выбранных узлов  
TEMP=T_INP1  
NSEL,S,LOC,Y,0 ! Выбор всех узлов с координатой Y=0  
NSEL,R,LOC,X,WL,WL-WWL  
D,ALL,TEMP,T_INP2 ! Задать для всех выбранных узлов  
TEMP=T_INP2  
NSEL,S,LOC,Y,HL ! Выбор всех узлов с координатой Y=HL  
NSEL,R,LOC,X,0,WWL  
SF,ALL,CONV,H_F,T_EXT ! Конвективный теплообмен в выбранных узлах  
NSEL,ALL ! Вернуться к выбору всех узлов модели  
  
SOLVE ! Решить СЛАУ МКЭ  
SAVE,Heat_Letter_R,db ! Сохранить базу ANSYS задачи под именем  
Heat_Letter_R.db  
FINISH
```

Отметим, что здесь по команде **SAVE,Heat_Letter_R,db** твердотельная и конечно-элементная модели задачи записываются в файл формата базы данных ANSYS *.db, и этому файлу назначено имя Heat_Letter_R.db.

Выведем получающуюся конечно-элементную сетку с граничными условиями (рис. 4). Для этого в интерактивном режиме следует выполнить: Plot->Elements, PlotCtrls->Symbols->Отметить All Applied BCs, в пункте Surface Load Symbols для отображения условий конвективного теплообмена задать вывод Convect FilmCoef:




Рис. 4. Конечно-элементная сетка с граничными условиями

Остается посмотреть результаты, что можно сделать в пост-процессоре POST1:

/POST1

! Команды, управляющие форматом графического вывода

/SHOW,WIN32C

/TRIAD,OFF ! Не показывать начала координат и осей

/PLOPTS,INFO,2 ! Использовать формат вывода Auto-legend для подписей

/PLOPTS,LEG2,OFF

/PLOPTS,LOGO,OFF ! Логотип ANSYS не показывать в графическом виде

/PLOPTS,FRAME,OFF ! Не показывать рамку

/PLOPTS,TITLE,OFF ! Не показывать заголовков

/PLOPTS,MINM,ON

/PLOPTS,DATE,OFF ! Не показывать дату

! Для показа распределений величин использовать 14 градаций уровней

/CONT,,14

! Инвертирование фона с черного на белый

/RGB,INDEX,100,100,100,0

/RGB,INDEX,0,0,0,15

/PBC,TEMP,,1 ! Отображать граничные условия для температуры
/PSF,CONV,НСOEF,2 ! Отображать стрелками граничные условия с коэффициентом теплообмена
PLNSOL,TEMP ! Показать поле температуры

Здесь вначале записано большое число необязательных команд, управляющих форматами вывода картинок. Собственно вывод распределения поля температуры осуществляется по последней команде **PLNSOL,TEMP**, которая дает результат, показанный на рис. 5. Для вывода распределение температуры в интерактивном режиме ANSYS следует выполнить вход в General Postprocessor: General Postproc->Plot Results->Contour Plot->Nodal Solu->DOF Solution->Nodal Temperature.




Рис. 5. Распределение температуры (с изображением граничных условий)

Из рис. 5 находим, что на верхней границе области с условием конвективного теплообмена получают минимальные значения температуры θ_h , приближенно равные 51°C .

В постпроцессоре **/POST1** можно вывести и другие характеристики решения, например, модуль вектора потока тепла. Для этого можно воспользоваться командами:

/PBC,TEMP,,0 ! Не показывать граничные условия для температуры
/PSF,CONV,НСOEF,0 ! Не показывать граничные условия с коэффициентом теплообмена
PLNSOL,TF,SUM ! Показать модуль вектора теплового потока

Или в интерактивном режиме: General Postproc->Plot Results->Contour Plot->Nodal Solu->Thermal flux->Thermal flux vector sum.

В результате получим картинку распределений значений модуля вектора потока тепла (рис. 6)




Рис. 6. Распределение модуля вектора потока тепла

Как видно из рис. 6, вектор потока тепла имеет экстремумы в угловых точках.

На рис. 7 показано распределение вектора потока тепла (General Postproc->Plot Results->Vector Plot->Predefined->Flux&gradient->Thermal flux TF.




Рис. 7. Распределение вектора потока тепла

Суммарный командный файл на языке APDL ANSYS называется Th2LS_1 (R).inp

Варианты заданий

Требуется рассчитать поле температур, используя аналогичные физические входные данные, что и рассмотренном примере, но для других областей, соответствующих буквам из таблицы 1. Геометрические размеры областей надо придумать самостоятельно в диапазонах значений, аналогичных рассмотренному выше примеру. Для построения букв использовать дуги эллипсов. Проведите расчеты в ANSYS. Проанализируйте результаты и оформите отчет.

Требования к отчету.

Отчет должен содержать ФИО студентов полное описание задачи, а также результаты, полученные с помощью конечно-элементного комплекса ANSYS в интерактивном и командном режимах.

В качестве результатов расчетов приведите:

- Конечно-элементную сетку с граничными условиями
- картину распределения температуры
- картину распределения вектора потока тепла
- картину распределения модуля вектора потока тепла

Таблица 1

№ задания	Вид области	ФИО студента
1.	Б	
2.	В	
3.	З	
4.	О	
5.	Р	
6.	С	
7.	У	
8.	Ф	
9.	Ц	
10.	Э	
11.	Ю	
12.	Я	
13.	D	
14.	G	
15.	J	
16.	Q	
17.	S	
18.	U	
19.	Ω	

20.	€	
21.	α	
22.	β	